International Hydrographic Review, Monaco, LXII (2), July 1985

A STEP FORWARD IN THE METHODS OF GETTING A FIX AT SEA

by Lt. (RNA) (H) Vicente MAESTRE GIMENO (*)

ABSTRACT

This paper deals with a method of computing a set of astronomical observations to two or more stars without requiring a knowledge of the approximate geographical position of the observer in order to obtain a fix. Programmes for use with the Hewlett-Packard HP 41/CV calculator and in BASIC language are included. The author also describes an extended application of the method to processing observations made by "radio sextants" to provide twenty-four-hour continuous positioning offshore without recourse to radio, satellite, or inertial positioning systems.

***

It is a fact well known among seamen that methods of getting an astronomical position at sea require the concurrence of dead reckoning. Seamen always know their estimated position, even if in some cases it is a very approximate one.

On the other hand, when a navigator carries out an astronomical observation, measuring the angle of altitude of two celestial bodies of his choice, the circles of altitude intersect at the two extreme points A and B, which represent an ambiguity with regard to the true geometrical location of his observation. However, this ambiguity does not exist in practice because, as we have said, the observer always has a dead reckoning available.

Nevertheless, if we consider position finding from a different point of view, the navigator can determine his fix without the concurrence of dead reckoning.

To do this we start from the navigational triangle, where (see Figures 2 and 3) :

PnPs, the line of the poles
ZZ′, the observer's zenith nadir line
QQ′, the equator

(*) Instituto Hidrografico de la Marina, Calle Tolosa Latour No. 1, CADIZ, Spain.

HH′, the observer's celestial horizon
A, the position of the celestial body, parallactic angle
Z, the azimuth of the celestial body
h, the altitude of the celestial body
φ, the observer's latitude
d, the declination of the celestial body
LHA, the local hour angle.

Applying to the spherical triangle APnZ the law of cosines, we obtain :

$$\cos(90^\circ - h) = \cos(90^\circ - \varphi)\cos(90^\circ - d) + \sin(90^\circ - \varphi)\sin(90^\circ - d)\cos LHA$$

$$\sin h_e = \sin\varphi_e \sin d + \cos\varphi_2 \cos d \cos LHA_e \quad (1)$$

(suffix e indicates the value is estimated).



FIG. 1



FIG. 2



FIG. 3

The equation (1) gives us a line of position on observing the altitude of a heavenly body "h", and we can obtain from the formula the computed or estimated altitude and "h_e". If we then compare these two altitudes, we will get an increment of altitude "Δ h", which with its azimuth determines the corresponding line of position. Its intersection with another line of position, from the observation of another celestial body, will provide the fix required.

Up to this point, we are still dependent upon dead reckoning.

Now if we establish a system of equations such as :

$\sin h_1 = \sin \varphi \sin d_1 + \cos \varphi \cos d_1 \cos LHA_1$ for celestial body A 1
$\sin h_2 = \sin \varphi \sin d_2 + \cos \varphi \cos d_2 \cos LHA_2$ for celestial body A 2

which converts to :

$$\sin h_1 \sin \varphi \sin d_2 + \sin h_1 \cos \varphi \cos d_2 \cos LHA_2 = \sin h_2 \sin \varphi \sin d_1 + \sin h_2 \cos \varphi \cos d_1 \cos LHA_1$$

and, dividing by $\cos \varphi$:

$$\sin h_1 \tan \varphi \sin d_2 + \sin h_1 \cos d_2 \cos LHA_2 = \sin h_2 \tan \varphi \sin d_1 + \sin h_2 \cos d_1 \cos LHA_1$$

and, taking out $\tan \varphi$ as common factor :

$$\tan \varphi (\sin h_1 \sin d_2 - \sin h_2 \sin d_1) = \sin h_2 \cos d_1 \cos LHA_1 - \sin h_1 \cos d_2 \cos LHA_2$$

and resolving $\tan \varphi$:

$$\tan \varphi = \frac{\sin h_2 \cos d_1 \cos LHA_1 - \sin h_1 \cos d_2 \cos LHA_2}{\sin h_1 \sin d_2 - \sin h_2 \sin d_1} \quad (2)$$

We still have the latitude as a function of the longitude. If we apply the same formula to a third celestial body of altitude "h_3", declination "d_3" and local hour angle "LHA_3" and we go on with the same transformations we will have :

$$\tan \varphi = \frac{\sin h_3 \cos d_1 \cos LHA_1 - \sin h_1 \cos d_3 \cos LHA_3}{\sin h_1 \sin d_3 - \sin h_3 \sin d_1} \quad (3)$$

If we equalize formulas (2) and (3) and go on with the same conversions we will get :

$$\frac{\sin h_2 \cos d_1 \cos LHA_1 - \sin h_1 \cos d_2 \cos LHA_2}{\sin h_1 \sin d_2 - \sin h_2 \sin d_1} = \frac{\sin h_3 \cos d_1 \cos LHA_1 - \sin h_1 \cos d_3 \cos LHA_3}{\sin h_1 \sin d_3 - \sin h_3 \sin d_1}$$

and this expression expands to :

$$(\sin h_2 \cos d_1 \cos LHA_1 - \sin h_1 \cos d_2 \cos LHA_2)(\sin h_1 \sin d_3 - \sin h_3 \sin d_1) = (\sin h_3 \cos d_1 \cos LHA_1 - \sin h_1 \cos d_3 \cos LHA_3)(\sin h_1 \sin d_2 - \sin h_2 \sin d_1)$$

or :

$$\begin{aligned}
&\sin h_2 \cos d_1 \cos LHA_1 \sin h_1 \sin d_3 - \sin h_2 \cos d_1 \cos LHA_1 \sin h_3 \sin d_1 \\
&- \sin h_1 \cos d_2 \cos LHA_2 \sin h_1 \sin d_3 + \sin h_1 \cos d_2 \cos LHA_2 \sin h_3 \sin d_1 \\
&= \sin h_3 \cos d_1 \cos LHA_1 \sin h_1 \sin d_2 - \sin h_3 \cos d_1 \cos LHA_1 \sin h_2 \sin d_1 \\
&- \sin h_1 \cos d_3 \cos LHA_3 \sin h_1 \sin d_2 + \sin h_1 \cos d_3 \cos LHA_3 \sin h_2 \sin d_1
\end{aligned}$$

and, taking out cos LHA_1, cos LHA_2 and cos LHA_3 as common factors :

$$\cos LHA_1 (\sin h_2 \cos d_1 \sin h_1 \sin d_3 - \sin h_2 \cos d_1 \sin h_3 \sin d_1$$
$$- \sin h_3 \cos d_1 \sin h_1 \sin d_2 + \sin h_3 \cos d_1 \sin h_2 \sin d_1)$$
$$+ \cos LHA_2 (\sin h_1 \cos d_2 \sin h_3 \sin d_1 - \sin h_1 \cos d_2 \sin h_1 \sin d_3)$$
$$= \cos LHA_3 (\sin h_1 \cos d_3 \sin h_2 \sin d_1 - \sin h_1 \cos d_3 \sin h_1 \sin d_2)$$

then, cancelling like terms and simplifying we have :

$$\cos LHA_1 \cos d_1 \sin h_1 (\sin h_2 \sin d_3 - \sin h_3 \sin d_2)$$
$$+ \cos LHA_2 \cos d_2 \sin h_1 (\sin h_3 \sin d_1 - \sin h_1 \sin d_3)$$
$$= \cos LHA_3 \cos d_3 \sin h_1 (\sin h_2 \sin d_1 - \sin h_1 \sin d_2)$$

To facilitate the introduction of these formulae in a computer and/or programmable calculator, we enter the following variables :

$$A = \cos d_1 \sin h_1 (\sin h_2 \sin d_3 - \sin h_3 \sin d_2)$$
$$B = \cos d_2 \sin h_1 (\sin h_3 \sin d_1 - \sin h_1 \sin d_3)$$
$$C = \cos d_3 \sin h_1 (\sin h_2 \sin d_1 - \sin h_1 \sin d_2)$$

and subsequently :

$$A \cos LHA_1 + B \cos LHA_2 = C \cos LHA_3$$

where :

$$LHA_1 = GHA_1 + L$$
$$LHA_2 = GHA_2 + L$$
$$LHA_3 = GHA_3 + L$$

GHA_1, GHA_2 and GHA_3 are the hour angles of the celestial bodies at Greenwich, and "L" the longitude, which at first we consider as positive for east longitudes, and the same formula with its sign solution will indicate for us the west longitudes.

If we follow the process of conversions and make the corresponding substitutions we obtain :

$$A \cos (GHA_1 + L) + B \cos (GHA_2 + L) = C \cos (GHA_3 + L)$$

and, expanding the cosines :

$$A (\cos GHA_1 \cos L - \sin GHA_1 \sin L)$$
$$+ B (\cos GHA_2 \cos L - \sin GHA_2 \sin L)$$
$$= C (\cos GHA_3 \cos L - \sin GHA_3 \sin L)$$

$$A \cos GHA_1 \cos L - A \sin GHA_1 \sin L$$
$$+ B \cos GHA_2 \cos L - B \sin GHA_2 \sin L$$
$$= C \cos GHA_3 \cos L - C \sin GHA_3 \sin L$$

then, taking out cos L and sin L as common factors :

$$\cos L (A \cos GHA_1 + B \cos GHA_2 - C \cos GHA_3)$$
$$= \sin L (A \sin GHA_1 + B \sin GHA_2 - C \sin GHA_3)$$

and, dividing by cos L :

$$A \cos GHA_1 + B \cos GHA_2 - C \cos GHA_3$$
$$= \tan L (A \sin GHA_1 + B \sin GHA_2 - C \sin GHA_3)$$

or :

$$\tan L = \frac{A \cos GHA_1 + B \cos GHA_2 - C \cos GHA_3}{A \sin GHA_1 + B \sin GHA_2 - C \sin GHA_3}$$

In this way we obtain the longitude and thus the latitude, using known data : the hour angles at Greenwich GHA_1, GHA_2 and GHA_3; the declinations d_1, d_2 and d_3, (all data obtainable from the Nautical Almanac using the universal time (U.T.) of observation) and the three altitudes observed by the navigator (h_1, h_2 and h_3).

We have seen that for calculating the observer's geometrical location by applying the attained formulae, it is necessary that the observation of those three celestial bodies be taken simultaneously, that is to say, at the same universal time U.T. This is impossible for only one observer and would require the collaboration of three observers. On the other hand observations are not so precise that they allow us to attain a common geometrical location for the three corresponding circles of altitude, particularly since the navigator observing three celestial bodies will usually give more weighting to one line of position than to others, depending on conditions of the horizon, the collimation, meteorological states, etc.

The latitude can be obtained by any of the following formulae :

$$\tan \text{lat.} = \frac{\sin h_2 \cos d_1 \cos LHA_1 - \sin h_1 \cos d_2 \cos LHA_2}{\sin h_1 \sin d_2 - \sin h_2 \sin d_1}$$

for stars A 1 and A 2

or else :

$$\tan \text{lat.} = \frac{\sin h_2 \cos d_1 \cos LHA_1 - \sin h_1 \cos d_2 \cos LHA_2}{\sin d_2 \cos d_1 \cos LHA_1 - \sin d_1 \cos d_2 \cos LHA_2}$$

which can be deduced easily from the previous formulae, with :

$$\tan L = \frac{A \cos GHA_1 + B \cos GHA_2 - C \cos GHA_3}{A \sin GHA_1 + B \sin GHA_2 - C \sin GHA_3}$$

where :

$$\begin{aligned} A &= \cos d_1 (\sin h_3 \sin d_2 - \sin h_2 \sin d_3) \\ B &= \cos d_2 (\sin h_1 \sin d_3 - \sin h_3 \sin d_1) \\ C &= \cos d_3 (\sin h_1 \sin d_2 - 2 \sin h_2 \sin d_1) \end{aligned}$$

$$\tan \text{lat.} = \frac{\sin h_3 \cos d_1 \cos LHA_1 - \sin h_1 \cos d_3 \cos LHA_3}{\sin h_1 \sin d_3 - \sin h_3 \sin d_1}$$

for stars A 1 and A 3

or using the above sinus formula :

$$\tan \text{lat.} = \frac{\sin h_2 \cos d_3 \cos LHA_3 - \sin h_3 \cos d_2 \cos LHA_2}{\sin h_3 \sin d_2 - \sin h_2 \sin d_3}$$

for stars A 2 and A 3.

In any case, if we were in the situation where the observations were taken by a single observer, he could use the obtained position from the above formulae as an estimated one and subsequently rework his position as a normal running fix, or else as follows :

Let us suppose on board a ship sailing course AB (see Fig. 4), the celestial body A 1 is observed to obtain the line of position CD or EF or even GH; this line of position only indicates which are the geometrical points of the position. A short interval of time later, another celestial body A 2 is observed, getting the line of position IJ or KL or even MN.

These lines of position do not necessarily give geometrical points such as "O" or "P" on the course, for example, but, as is obvious, this could be on any of the



FIG. 4

lines of position in the figure, in agreement with the astronomical observations taken.

Now the problem which poses itself is how to find out which geometrical point corresponds to the fix we are trying to get.

Let us suppose that as a consequence of the observation taken, the lines of position that we have to work on are GH for celestial body A 1 and IJ for A 2, outside the course AB, for example.

We then transfer the line of position IJ to the position of the first one GH.

Observing the triangle A′TB′, right angle in T (see Fig. 5), we obtain :

$$\Delta h = D \cos (C - Z_2)$$

where :

- Δh increment of altitude
- D distance navigated
- C track followed by the ship
- Z_2 azimuth of star A 2.

We see that the increment of altitude "Δh" is the correction to be applied to the height of star A 2, so as to run it from position 2 to 1.

$$\Delta h = D \cos (180^\circ - (C - Z_2)) = - D \cos (C - Z_2)$$

and in general $\Delta h = D \cos (C - Z)$.



FIG. 5

This formula gives us the increment of altitude and its corresponding sign, and it is a general one which can be verified for the four quadrants, for both azimuth and course, whenever course and azimuth are taken in a clockwise direction, i.e. from the north to the east, as is usual in nautical astronomy.

We can follow the same reasoning for a third celestial body and work out the fix from the last observed position.

To illustrate with an example :

A navigator sails course S 10 W at a speed of 30 knots, on the 4th July 1984 and observes the following stars :

U.T. = 20 h 55 m 49 s	DENEBOLA	h_1 = 60°05.2′	above the celestial horizon.
U.T. = 20 h 57 m 45 s	SPICA	h_2 = 53°24.1′	
U.T. = 20 h 59 m 50 s	SABIK	h_3 = 27°00.0′	

What is his fix ?

The first calculation we undertake is that of obtaining the fix using the heights just as they are and correcting them later on for transference of the lines of position of DENEBOLA and SPICA to that of SABIK.

	DENEBOLA	SPICA	SABIK
	U.T. = 20 h 55 m 49 s	U.T. = 20 h 57 m 45 s	U.T. = 20 h 59 m 50 s
	d_1 = 14°39.7′	−11°04.8′	−15°42.4′
Aries H.A.	= 223°00.4′	223°00.4′	223°00.4′
Correction	= 13°59.5′	14°28.6′	15°00.0′
Correc. Aries H.A.	= 236°59.9′	237°29.0′	238°00.4′
SHA	= 182°56.3′	158°54.6′	102°37.7′
GHA_1	= 419°56.2′	GHA_2 = 396°23.6′	GHA_3 = 340°38.1′



FIG. 6. — Flow Chart.

In the programmable calculator HP 41/CV in which previously we have loaded the program by means of the corresponding cards, we enter the following data :

h_1 = 60.052	d_1 = 14.397	GHA_1 = 419.562
h_2 = 53.241	d_2 = − 11.048	GHA_2 = 396.236
h_3 = 27.000	d_3 = − 15.424	GHA_3 = 340.381

Once these data have been entered, following the program, we get :

HP 41/CV Print-out

```
          XEQ "LATLON"
ALTURA ASTROS  ?
a1  ENTER ↑
a2  ENTER ↑
a3   R/S
         60.052 ENTER ↑
         53.241 ENTER ↑
         27.000     RUN

DECLINACION ASTROS  ?
d1  ENTER ↑
d2  ENTER ↑
d3  R/S
         14.397 ENTER ↑
        -11.048 ENTER ↑
        -15.424     RUN

HORARIO ASTROS  ?
H*G1  ENTER ↑
H*G2  ENTER ↑
H*G3   R/S
         419.562 ENTER ↑
         396.236 ENTER ↑
         340.381     RUN

HEMISFERIO NORTE O SUR ?
              N    S R/S
N                    RUN

LAT. 24.° 59.0'N

LON  29.° 59.7' W

          24.9833 STO 00
          14.6617 STO 01
         389.9417 STO 02
              XEQ "ALT"
            104.5    ***
                  STO 00
           360.0 ENTER ↑
                  RCL 00
                       -
           255.5      ***
```

Lat. 24°59.0′ N Lon 29°59.7′ W

This is a first approximation of the fix we want to obtain: with this E.P. (estimated position) we can calculate the azimuths of the stars DENEBOLA and SPICA.

DENEBOLA			SPICA		
GHA_1	=	419°56.2′	GHA_2	=	396°23.6′
Lon	=	29°59.7′ −	Lon	=	29°59.7′ −
LHA	=	389°56.5′	LHA	=	366°23.9′

We enter in HP 41/CV calculator :

For DENEBOLA :

Lat	=	24°59′	=	24.9833°
d_1	=	14°39.7′	=	14.6617°
LHA	=	389°56.5′	=	389.9417°

getting azimuth 255.5°.

```
24.9833 STO ØØ
-11.Ø8ØØ STO Ø1
366.3983 STO Ø2
      XEQ "ALT"
  169.4      ***
          STO ØØ
  36Ø.Ø  ENTER↑
          RCL ØØ
               -
  19Ø.6      ***
```

For SPICA :

Lat	=	24°59′	=	24.9833°
d_2	=	− 11°04.8′	=	− 11.0800°
LHA	=	366°23.9′	=	366.3983°

getting azimuth 190.6°.

```
ALTURA ASTROS  ?
a1  ENTER↑
a2  ENTER↑
a3   R/S
          6Ø.Ø6ØØ ENTER↑
          53.251Ø ENTER↑
          27.ØØØØ    RUN

DECLINACION ASTROS  ?
d1  ENTER↑
d2  ENTER↑
d3  R/S
           14.397Ø ENTER↑
          -11.Ø48Ø ENTER↑
          -15.424Ø    RUN

HORARIO ASTROS  ?
H*G1  ENTER↑
H*G2  ENTER↑
H*G3   R/S
          419.562Ø ENTER↑
          396.236Ø ENTER↑
          34Ø.381Ø    RUN

HEMISFERIO NORTE O SUR ?
                N    S R/S
N                      RUN

LAT.  24.° 58.Ø'N

LON   3Ø.° Ø.4' W
```

Now we can correct the heights of DENEBOLA and SPICA using the formula $\Delta h = D \cos (C - Z)$, to get the fix at U.T. = 20 h 59 m 50 s.

DENEBOLA		SPICA	
Course	= 190°	Course	= 190°
Azimuth	= 255.5°	Azimuth	= 190.6°
C − Z	= 065.5° −	C − Z	= 000.6° −
U.T.	= 20 h 59 m 50 s	U.T.	= 20 h 59 m 50 s
U.T.	= 20 h 55 m 49 s	U.T.	= 20 h 57 m 45 s
t	= 4 m 01 s	t	= 2 m 05 s
D	= 2′	D	= 1′
Δ h	= + 0.8′	Δ h	= + 1′
h	= 60°05.2′	h	= 53°24.1′
Δ h	= 0.8′ +	Δ h	= 1.0′ +
correc. h_1	= 60°06.0′	correc. h_2	= 53°25.1′

We enter the data once again in the HP 41/CV calculator :

h_1 = 60.060	d_1 = 14.397	GHA_1 = 419.562
h_2 = 53.251	d_2 = − 11.048	GHA_2 = 396.236
h_3 = 27.000	d_3 = − 15.424	GHA_3 = 340.381

thereby getting the desired fix :

Lat. 24°58.0′N Lon. 30°00.4′W

To check this position we work it out as a normal running fix, starting from the approximate position Lat. 24°59.0′N Lon. 29°59.7′W and getting :

Lat. 24°58.2′N Lon. 30°00.2′W

Using the attached BASIC program will lead us to the same result.

The primary advantage (among others) of this method is its use with the radio-sextant.

It is possible at present to get a fix without using the classic sextant (ignoring radio-positioning systems and satellites). To obtain the angle of altitude of a heavenly body, it is possible to match it by means of radio-sextants, since heavenly bodies emit radiations on their own frequencies showing a visible spectrum or radio wave.

This system is used to calibrate the position of inertial systems of navigation; if we have three receivers available to measure simultaneously the corresponding radiation of groups of three stars, it would always be possible to have available a position of high precision (even for research vessels), whatever meteorological conditions are present, throughout the 24 hours of a day, since these radiations are also received in the day time. In this way it would be feasible to obtain a fix independent from onshore radioelectric stations or satellites.

These formulae can also be applied to land observations, since the attainment of an astronomical point of a geodetic triangular net would not be restricted to the observation of circumpolar stars, that is to say, stars having the same altitude or else the same local hour angle at both sides of the meridian. This system has the advantage that the observer has very easy access to his means of observations, since he can choose the most suitable stars to give the very best intersections.

This paper is intended at present for yachtsmen; however, when electronic development is completed, the principles involved could be a useful tool for navigators, surveyors and also for military applications.

HP 41/CV Calculator Program

```
PRP "LATLON"

Ø1*LBL "LATLON"
Ø2 "ALTURA"
Ø3 ACA
Ø4 "ASTROS"
Ø5 ASTO 29
Ø6 ACA
Ø7 SF 12
Ø8 " ?"
Ø9 ASTO 28
10 ACA
11 CF 12
12 PRBUF
13 "a1  ENTER↑"
14 PRA
15 "a2  ENTER↑"
16 PRA
17 "a3   R/S"
18 PRA
19 TONE 9
2Ø STOP
21 ADV
22 STO Ø2
23 RDN
24 STO Ø1
25 RDN
26 STO ØØ
27 "DECLINACION"
28 ACA
29 CLA
3Ø ARCL 29
31 ACA
32 CLA
33 ARCL 28
34 SF 12
35 ACA
36 CF 12
37 PRBUF
38 "d1  ENTER↑"
39 PRA
4Ø "d2  ENTER↑"
41 PRA
42 "d3  R/S"
43 PRA
44 TONE 7
45 STOP
46 ADV
47 STO Ø5
48 RDN
49 STO Ø4
5Ø RDN
51 STO Ø3
52 "HORARIO"
53 ACA
54 CLA
55 ARCL 29
56 ACA
57 CLA
58 ARCL 28
59 SF 12
6Ø ACA
61 CF 12
62 PRBUF
63 "H*G1  ENTER↑"
64 PRA
65 "H*G2  ENTER↑"
66 PRA
67 "H*G3     R/S"
68 PRA
69 TONE 5
7Ø STOP
71 STO Ø8
72 RDN
73 STO Ø7
74 RDN
75 XEQ "BES"
76 STO Ø6
77 RCL Ø8
78 XEQ "BES"
79 STO Ø8
8Ø RCL Ø2
81 XEQ "BES"
82 STO Ø2
83 RCL Ø3
84 XEQ "BES"
85 STO Ø3
86 RCL Ø4
87 XEQ "BES"
88 STO Ø4
89 RCL Ø5
9Ø XEQ "BES"
91 STO Ø5
92 RCL Ø7
93 XEQ "BES"
94 STO Ø7
95 RCL ØØ
96 XEQ "BES"
97 STO ØØ
98 RCL Ø1
99 XEQ "BES"
1ØØ STO Ø1
1Ø1 SIN
1Ø2 STO Ø9
1Ø3 RCL Ø5
1Ø4 SIN
1Ø5 STO 1Ø
1Ø6 *
1Ø7 RCL Ø2
1Ø8 SIN
1Ø9 STO 11
11Ø RCL Ø4
111 SIN
112 STO 12
113 *
114 -
115 RCL ØØ
116 SIN
117 STO 13
118 RCL Ø3
119 COS
12Ø STO 14
121 *
122 *
123 STO 15
124 RCL 11
125 RCL Ø3
126 SIN
127 STO 16
128 *
129 RCL 13
13Ø RCL 1Ø
131 *
132 -
133 RCL 13
134 RCL Ø4
135 COS
136 STO 17
137 *
138 *
139 STO 18
14Ø RCL Ø9
141 RCL 16
142 *
143 RCL 13
144 RCL 12
145 *
146 -
147 RCL 13
148 RCL Ø5
149 COS
15Ø STO 19
151 *
152 *
153 STO 2Ø
154 RCL 15
155 RCL Ø6
156 COS
157 *
158 RCL 18
159 RCL Ø7
16Ø COS
161 *
162 +
163 RCL 2Ø
164 RCL Ø8
165 COS
166 *
167 -
168 RCL 15
169 RCL Ø6
17Ø SIN
171 *
172 RCL 18
173 RCL Ø7
174 SIN
175 *
176 +
177 RCL 2Ø
178 RCL Ø8
179 SIN
18Ø *
181 -
182 /
183 ATAN
184 STO 21
185 RCL Ø6
186 +
187 COS
188 STO 22
189 RCL 14
19Ø *
191 RCL Ø9
192 *
193 RCL 21
194 RCL Ø7
195 +
196 COS
197 STO 23
198 RCL 17
199 *
2ØØ RCL 13
2Ø1 *
2Ø2 -
2Ø3 RCL 13
2Ø4 RCL 12
2Ø5 *
2Ø6 RCL Ø9
2Ø7 RCL 16
2Ø8 *
2Ø9 -
21Ø /
211 ATAN
212 STO 25
213 ADV
214 CLA
215 "HEMISFERIO NORT"
216 ACA
217 "E O SUR ?"
218 ACA
219 "N     S R/S"
22Ø ACA
221 ADV
222 "N"
223 ASTO Y
224 CLA
225 AON
226 TONE 5
227 STOP
228 ASTO X
229 ADV
23Ø AOFF
231 X=Y?
232 GTO "N"

233*LBL "S"
234 "S"
235 ASTO 33
236 RCL 25
237 XEQ "T"
238 XEQ "Z"
239 STOP

24Ø*LBL "T"
241 X<Ø?
242 GTO Ø4
243 STO 36
244 RCL 21
245 X<Ø?
246 GTO "X"
247 18Ø
248 -
249 STO 32

25Ø*LBL Ø3
251 RCL 36
252 "LAT."
253 ACA
254 CLA
255 "t"
256 ACA
257 CLA
258 XEQ J
259 ARCL 33
26Ø ACA
261 PRBUF
262 RTN

263*LBL Ø4
264 ABS
265 STO 36
266 RCL 21
267 STO 32
268 GTO Ø3

269*LBL "Z"
27Ø ADV
271 RCL 32
272 XEQ I
273 "LON"
274 ACA
275 XEQ J
276 CLA
277 ARCL 3Ø
278 ACA
279 PRBUF
28Ø BEEP
281 RTN

282*LBL "X"
283 18Ø
284 +
285 STO 32
286 GTO Ø3

287*LBL I
288 X<Ø?
289 GTO Ø1
29Ø "E"
291 ASTO 3Ø
292 RTN

293*LBL Ø1
294 ABS
295 STO 32
296 "W"
297 ASTO 3Ø
298 RTN

299*LBL F
3ØØ 1
3Ø1 ST+ 34
3Ø2 Ø
3Ø3 STO 35
3Ø4 GTO G

3Ø5*LBL J
3Ø6 ENTER↑
3Ø7 INT
3Ø8 STO 34
3Ø9 -
31Ø 6Ø
311 *
312 ABS
313 FIX 1
314 RND
315 6Ø
316 X=Y?
317 GTO F
318 X<>Y
319 STO 35

32Ø*LBL G
321 RCL 34
322 FIX Ø
323 ACX
324 7
325 BLDSPEC
326 5
327 BLDSPEC
328 5
329 BLDSPEC
33Ø 7
331 BLDSPEC
332 Ø
333 BLDSPEC
334 Ø
335 BLDSPEC
336 Ø
337 BLDSPEC
338 ACSPEC
339 RCL 35
34Ø FIX 1
341 ACX
342 39
343 ACCHR
344 RTN

345*LBL "N"
346 "N"
347 ASTO 33
348 RCL 25
349 XEQ "K"
35Ø XEQ "Z"
351 STOP

352*LBL "K"
353 X>Ø?
354 GTO "L"
355 ABS
356 STO 36
357 XEQ "T"
358 RTN

359*LBL "L"
36Ø STO 36
361 RCL 21
362 STO 32
363 XEQ Ø3
364 RTN

365*LBL "BES"
366 INT
367 LASTX
368 FRC
369 .6
37Ø /
371 +
372 RTN
373 END
```

This program gives you the azimuths and altitudes as well, to run the fix as a normal one.

To get azimuths you substract the value provided by the calculator from 360° if sin LHA > 0 or else take azimuth just as it is in the clockwise direction. See pages 128 (bottom), 129 (top).

```
PRP  "ALT"

Ø1♦LBL  "ALT"
 Ø2 RCL ØØ
 Ø3 SIN
 Ø4 RCL Ø1
 Ø5 SIN
 Ø6 STO Ø3
 Ø7 *
 Ø8 RCL ØØ
 Ø9 COS
 1Ø STO Ø4
 11 RCL Ø1
 12 COS
 13 *
 14 RCL Ø2
 15 COS
 16 *
 17 +
 18 ASIN
 19 PRX
 2Ø STO Ø5
 21 RCL Ø3
 22 RCL Ø4
 23 RCL Ø5
 24 COS
 25 *
 26 /
 27 RCL ØØ
 28 TAN
 29 RCL Ø5
 3Ø TAN
 31 *
 32 -
 33 ACOS
 34 7PRTX
 35 END
XEQ "LATLON"
```

```
ALTURA ASTROS  ?
a1  ENTER↑
a2  ENTER↑
a3   R/S
         6Ø.86Ø ENTER↑
         53.251 ENTER↑
         27.ØØØ    RUN
```

Latlon Latitude, longitude
a1, a2, a3 stars' altitudes (heights)

```
DECLINACION ASTROS  ?
d1  ENTER↑
d2  ENTER↑
d3  R/S
          14.397 ENTER↑
         -11.Ø48 ENTER↑
         -15.424    RUN
```

d1, d2, d3 stars' declinations.

```
HORARIO ASTROS  ?
H*G1  ENTER↑
H*G2  ENTER↑
H*G3   R/S
         419.562 ENTER↑
         396.236 ENTER↑
         34Ø.381    RUN
```

H*G 1, H*G 2, H*G 3 Greenwich hour angles ?

```
HEMISFERIO NORTE O SUR ?
                N    S R/S
N                      RUN

LAT.  24.° 58.Ø'N

LON   3Ø.° Ø.4' W
```

Northern or southern hemisphere ?

BASIC PROGRAM

```
10      REM    Fix without dead reckoning
20      REM
30      REM
40      DIM A(3),D(3),H(3)
50      FOR I=1 TO 3
60        PRINT " STAR  HEIGHT ?          A";I,
70        INPUT A(I)
80        PRINT A(I)
90        P=A(I)
100       GOSUB 870
110       A(I)=P
120       PRINT " STAR DECLINATION ?       D";I,
130       INPUT D(I)
140       PRINT D(I)
150       P=D(I)
160       GOSUB 870
170       D(I)=P
180       PRINT " GREENWICH HOUR ANGLE ?  H";I,
190       INPUT H(I)
200       PRINT H(I)
210       P=H(I)
220       GOSUB 870
230       H(I)=P
240       PRINT
250     NEXT I
260     REM
270     REM      ********************
280     REM      *   Calculation    *
290     REM      ********************
300     REM
310     X=SIN(A(1))*COS(D(1))*(SIN(A(2))*SIN(D(3))-SIN(A(3))*SIN(D(2)))
320     Y=SIN(A(1))*COS(D(2))*(SIN(A(3))*SIN(D(1))-SIN(A(1))*SIN(D(3)))
330     Z=SIN(A(1))*COS(D(3))*(SIN(A(2))*SIN(D(1))-SIN(A(1))*SIN(D(2)))
340     Lon=(180/PI)*ATN((X*COS(H(1))+Y*COS(H(2))-Z*COS(H(3)))/(X*SIN(H(1))+Y*SIN
(H(2))-Z*SIN(H(3))))
350     FOR I=1 TO 3
360       H1(I)=H(I)+Lon*PI/180
370     NEXT I
380     Lat=(180/PI)*ATN((SIN(A(2))*COS(D(1))*COS(H1(1))-SIN(A(1))*COS(D(2))*COS(
H1(2)))/(SIN(A(1))*SIN(D(2))-SIN(A(2))*SIN(D(1))))
390     REM
400     REM      ******************
410     REM      *   Hemisphere   *
420     REM      ******************
430     REM
440     PRINT " NORTH OR SOUTH HEMISPHERE (N/S)? ",
450     INPUT J$
460     PRINT J$
470     IF J$="N" THEN GOTO 510
480     IF J$="S" THEN GOTO 620
490     GOTO 440
500     REM
510     REM           *************** NORTHERN  HEMISPHERE *******
520     IF Lat<0 THEN GOTO 540
530     GOTO 660
540     IF Lon<0 THEN GOTO 560
550     GOTO 580
560     Lon=Lon+180
570     GOTO 590
580     Lon=Lon-180
590     Lat=ABS(Lat)
600     GOTO 660
610     REM
620     REM           *************** SOUTHERN  HEMISPHERE **********
630     IF Lat>0 THEN GOTO 540
640     REM
```

```
650    REM      ***************
660    REM      *    Print    *
670    REM      ***************
680    REM
690    PRINT
700    PRINT
710    T=ABS(Lat)
720    N=ABS(Lon)
730    R=INT(T)
740    S=.1*INT(600*ABS(T-R))
750    G=INT(N)
760    M=.1*INT(600*ABS(N-G))
770    F$="E"
780    IF Lon<0 THEN F$="W"
790    PRINT USING """LATITUDE =  "",2Z,""°"",X,2Z.D,""'"",X,2A";R,S,J$
800    PRINT USING """LONGITUDE= "",3Z,""°"",X,2Z.D,""'"",X,2A";G,M,F$
810    STOP
820    REM
830    REM    ***********************
840    REM    *  Subrutine  radian  *
850    REM    ***********************
860    REM
870    IF P<0 THEN 900
880    P=(INT(P)+(P-INT(P))/.6)*PI/180
890    RETURN
900    P=-P
910    P=-(INT(P)+(P-INT(P))/.6)*PI/180
920    RETURN
930    END
```

PROGRAM RUN

```
STAR HEIGHT ?                A 1    60.06
STAR DECLINATION ?           D 1    14.397
GREENWICH HOUR ANGLE ?       H 1    419.562

STAR HEIGHT ?                A 2    53.251
STAR DECLINATION ?           D 2   -11.048
GREENWICH HOUR ANGLE ?       H 2    396.236

STAR HEIGHT ?                A 3    27
STAR DECLINATION ?           D 3   -15.424
GREENWICH HOUR ANGLE ?       H 3    340.381

NORTH OR SOUTH HEMISPHERE (N/S)?          N


LATITUDE =   24° 58.0' N
LONGITUDE= 030° 00.3' W
```

Note

The author of this work has prepared a second bilingual edition of his book "Nautical Astronomy — Fix without dead reckoning", where the errors that can arise with this system and their elimination are treated.

BIBLIOGRAPHY

American Practical Navigator, by Nathaniel Bowditch, LL.D. 1977 Edition.

Geodesia e Hidrografia, by Vicente Gandarias. 1961 Edition.

Astronomia de posición. Nuevo sistema de situación, by Vicente Maestre Gimeno. First Edition May 1984. Instituto Hidrográfico de la Marina, Cádiz, Spain.